
	Reason for change:
	Some corrections required in the implementation of Table 4.5A.2C.2.2-1: Procedure for UE-requested PDU session modification after the first S1 to N1 mode change.

1. When processing PDU SESSION MODIFICATION REQUEST, any information missing in the NAS message should be considered unchanged from the stored PDU session information, rather than changing to omit. For example, if the UE does not include a DNN name in the UL INFORMATION TRANSFER message carrying PDU SESSION MODIFICATION REQUEST, the current implementation will return  GSM_MobilityInfo_Type with DNN name set to omit and then PDU type will be updated. This can cause an IMS PDU session to be treated as internet type (or vice versa depending on PICS settings) which can lead to inconsistent DRB ID allocations.

2. Current TTCN implementation is setting p_ExpectedNumberOfNewPDUSessions = pc_noOf_PDUsSameConnection + pc_noOf_PDUsNewConnection. According to 38.508-1 table 4.5A.2C.1-1 the value should be based on the PDN pics parameters (used for S1 mode operation) not PDN. Also, the parameter  pc_noOf_PDNsSameConnection is defined as : “Number of UE requested PDN connection establishments after ATTACH in a new signaling connection”. This implies that the PDN connection which is made during the initial EPS attach procedure is not counted in these PICs parameters. So the total count of the PDN connection established in S1 mode (and hence the value for  p_ExpectedNumberOfNewPDUSessions) will be “pc_noOf_PDNsSameConnection + pc_noOf_PDNsNewConnection + 1”

3. In the current implementation the index used when retrieving a stored PDU SESSION MODIFICATION REQUEST is hardcoded to 1. This means the 1st received request will always be retrieved at each loop iteration.

4. In the case where a 2nd PDU SESSION MODIFICATION REQUEST has been received and stored while the 1st is being processed, the current function implementation, after sending PDU SESSION MODIFICATION COMMAND for the 1st request, will retrieve the stored PDU SESSION MODIFICATION REQUEST, increment the counter L to 2, and then deciding not to process it because it is waiting for PDU SESSION MODIFICATION COMPLETE. But because the counter L has already been incremented, after receiving PDU SESSION MODIFICATION COMPLETE, the function will then check L=N and so then does not trigger the processing of the 2nd request.


	
	

	Summary of change:
	1. When the existing PDU session information is found, use this to initialise  p_GSM_MobilityInfo, and then update fields only if they are present in the received messages. Acoid updating PDU Type for an existing PDU session.

2. Correct setting of p_ExpectedNumberOfNewPDUSessions to “pc_noOf_PDNsSameConnection + pc_noOf_PDNsNewConnection+1”

Note : An associated prose CR on 38.508-1 will be raised at RAN5#95.

3. Use the index “L+1” when calling f_NR5GC_MsgInDefault_GetOneNASMessageReq() to retrieve each stored PDU SESSION MODIFICATION REQUEST in turn.

4. Do not check for any stored messages if the variable v_WaitForComplete is set TRUE, since in this case the request will anyway not be processed.

	
	

	Consequences if not approved:
	A conformant UE operating in multi-PDN/PDU mode may fail NR5GC IRAT test cases

1. [bookmark: _Toc122434485][bookmark: _Toc100777898]Overview
This document lists all the changes needed to correct issues in the ATS iwd-TTCN3-B2020-09_D12wk12 related to the title of this CR.
Contact:	Shaun Harry
	shaun.harry@keysight.com
1. [bookmark: _Toc122434488][bookmark: _Toc295288959][bookmark: _Toc100777899]Corrections required

1. [bookmark: _Toc30685521][bookmark: _Toc84011980][bookmark: _Toc100777900]Change 1
	Function name
	f_Check_NG_PDUSessionModificationReq()

	Reason for change
	When processing PDU SESSION MODIFICATION REQUEST, any information missing in the NAS message should be considered unchanged from the stored PDU session information, rather than changing to omit. For example, if the UE does not include a DNN name in the UL INFORMATION TRANSFER message carrying PDU SESSION MODIFICATION REQUEST, the current implementation will return  GSM_MobilityInfo_Type with DNN name set to omit and then PDU type will be updated. This can cause an IMS PDU session to be treated as internet type (or vice versa depending on PICS settings) which can lead to inconsistent DRB ID allocations.

	Summary of change
	When the existing PDU session information is found, use this to initialise  p_GSM_MobilityInfo, and then update fields only if they are present in the received messages. Avoid updating PDU Type for an existing PDU session.

	TTCN module
	NG_NAS_TemplateFunctions.ttcn

	MCC160 Comment
	Accepted.  Also removed setting NSSAI to omit



Before Change
	function f_Check_NG_PDUSessionModificationReq( out GSM_MobilityInfo_Type p_GSM_MobilityInfo,
                                                 PDUSessionInfoList_Type p_PDUSessionGlobalInfo,
                                                 NG_NAS_UL_Pdu_Type p_ReceivedMsgs,
                                                 template (present) O1_Type p_PDU_SessionId := ?,
                                                 template (present) ProcedureTransactionIdentifier p_PTI := ?,
                                                 template NG_UE_SM_Cap p_NG_UE_SM_Cap := *,
                                                 template GMM_GSM_Cause p_Cause := *,
                                                 template MaxNumPacketFilters p_MaxNumPacketFilters := *,
                                                 template AlwaysOnPDUSessionReq p_AlwaysOnPDUSessionReq := *,
                                                 template IntegrityProtMaxDataRate p_IntegrityProtMaxDataRate := *,
                                                 template QoS_Rules p_QoS_Rules := *,
                                                 template QoSFlowDescr p_QoSFlowDescr := *,
                                                 template MappedEPSBearerContexts p_MappedEPSBearerContexts := *,
                                                 template ExtdProtocolConfigOptions p_ExtdPCO := *,
                                                 template NG_PDU_SessionId p_OldPDU_SessionId  := *,
                                                 template NG_Request_Type p_RequestType  := *,
                                                 template S_NSSAI_Type p_S_NSSAI  := *,
                                                 template DNN p_DNN  := *,
                                                 template AdditionalInformation p_AdditionalInfo  := *,
                                                 template MA_PDUSessionInfo p_MA_PDUSessionInfo := *,      // @sic R5s201387 Baseline Moving sic@
                                                 template ReleaseAssistanceInd p_ReleaseAssistanceInd := *,
                                                 template PortManagementInfoContainer p_PortManagementInfoContainer := *,
                                                 template IPHeaderCompressionConfig p_IPHeaderCompConfig := *,
                                                 template EthernetHeaderCompressConfig p_EthernetHeaderCompConfig := *) return boolean
  {
    var boolean v_Result := false;
    var boolean v_ExistingId := false;
    var NG_NAS_UL_Message_Type v_GSMMsg := p_ReceivedMsgs.PiggybackedPduList[0].Msg;
    var integer i;
    
    // Check NAS Transport msg first
    v_Result := f_Check_NG_ULNASTransport (p_ReceivedMsgs.Msg,
                                           cr_PayloadContainerType(tsc_PayloadContainerESMMsg), // @sic R5w190208 sic@
                                           p_PDU_SessionId,
                                           p_OldPDU_SessionId,
                                           p_RequestType,
                                           p_S_NSSAI,
                                           p_DNN,
                                           p_AdditionalInfo,
                                           p_MA_PDUSessionInfo,   // @sic R5s201387 Baseline Moving sic@
                                           p_ReleaseAssistanceInd);
    if (not v_Result) {
      return v_Result;
    } else if (match (v_GSMMsg, cr_NG_PDU_SESSION_MODIFICATION_REQUEST (cr_NG_PDU_SessionId(-, p_PDU_SessionId), // @sic R5s200564 sic@
                                                                        p_PTI,
                                                                        p_NG_UE_SM_Cap,
                                                                        p_Cause,
                                                                        p_MaxNumPacketFilters,
                                                                        p_AlwaysOnPDUSessionReq,
                                                                        p_IntegrityProtMaxDataRate,
                                                                        p_QoS_Rules,
                                                                        p_QoSFlowDescr,
                                                                        p_MappedEPSBearerContexts,
                                                                        p_ExtdPCO,
                                                                        p_PortManagementInfoContainer,  // @sic R5s201387 Baseline Moving sic@
                                                                        p_IPHeaderCompConfig,
                                                                        p_EthernetHeaderCompConfig))) {
      v_Result := true;
    } else {  // @sic R5s190944 sic@
      v_Result := false;
    }
    // Default message contents specifies this must match an existing session establishment id
    for (i := 0; i < lengthof(p_PDUSessionGlobalInfo); i := i+1) {
     if (match (v_GSMMsg.pdu_Session_Modification_Request.pduSessionId.sessionId, p_PDUSessionGlobalInfo[i].SessionId)) { // @sic R5s200564 sic@
       v_ExistingId := true;
     }
    }
    if (not v_ExistingId) {
      v_Result := v_ExistingId;
    }
    
    // Store info for this modification
    p_GSM_MobilityInfo.SessionId := v_GSMMsg.pdu_Session_Modification_Request.pduSessionId.sessionId; // @sic R5s200564 sic@
    p_GSM_MobilityInfo.PTI := v_GSMMsg.pdu_Session_Modification_Request.procedureTransactionIdentifier;
    p_GSM_MobilityInfo.SessionType := omit;
    p_GSM_MobilityInfo.EPS_Bearer := omit; // @sic R5s201533 sic@
    p_GSM_MobilityInfo.PdnIndex := omit; // @sic R5s201533 sic@
    
    if (ispresent (p_ReceivedMsgs.Msg.ul_Nas_Transport.dnn)) {
      p_GSM_MobilityInfo.DNN := p_ReceivedMsgs.Msg.ul_Nas_Transport.dnn;
    } else {
      p_GSM_MobilityInfo.DNN := omit;
    }
    p_GSM_MobilityInfo.PDUType := f_GetPDNTypeFromDNN(p_GSM_MobilityInfo.DNN);  // @sic R5-206296 sic@
    if (ispresent (v_GSMMsg.pdu_Session_Modification_Request.alwaysOnPDUSessionReq)) {
      p_GSM_MobilityInfo.AlwaysOn := v_GSMMsg.pdu_Session_Modification_Request.alwaysOnPDUSessionReq;
    } else {
      p_GSM_MobilityInfo.AlwaysOn := omit;
    }
    if (ispresent (v_GSMMsg.pdu_Session_Modification_Request.extdProtocolConfigurationOptions)) {
      p_GSM_MobilityInfo.PCO := v_GSMMsg.pdu_Session_Modification_Request.extdProtocolConfigurationOptions;
    } else {
      p_GSM_MobilityInfo.PCO := omit;
    }
    p_GSM_MobilityInfo.NSSAI := omit; // @sic R5s210408 sic@
    return v_Result;
  }



After Change
	function f_Check_NG_PDUSessionModificationReq( out GSM_MobilityInfo_Type p_GSM_MobilityInfo,
                                                 PDUSessionInfoList_Type p_PDUSessionGlobalInfo,
                                                 NG_NAS_UL_Pdu_Type p_ReceivedMsgs,
                                                 template (present) O1_Type p_PDU_SessionId := ?,
                                                 template (present) ProcedureTransactionIdentifier p_PTI := ?,
                                                 template NG_UE_SM_Cap p_NG_UE_SM_Cap := *,
                                                 template GMM_GSM_Cause p_Cause := *,
                                                 template MaxNumPacketFilters p_MaxNumPacketFilters := *,
                                                 template AlwaysOnPDUSessionReq p_AlwaysOnPDUSessionReq := *,
                                                 template IntegrityProtMaxDataRate p_IntegrityProtMaxDataRate := *,
                                                 template QoS_Rules p_QoS_Rules := *,
                                                 template QoSFlowDescr p_QoSFlowDescr := *,
                                                 template MappedEPSBearerContexts p_MappedEPSBearerContexts := *,
                                                 template ExtdProtocolConfigOptions p_ExtdPCO := *,
                                                 template NG_PDU_SessionId p_OldPDU_SessionId  := *,
                                                 template NG_Request_Type p_RequestType  := *,
                                                 template S_NSSAI_Type p_S_NSSAI  := *,
                                                 template DNN p_DNN  := *,
                                                 template AdditionalInformation p_AdditionalInfo  := *,
                                                 template MA_PDUSessionInfo p_MA_PDUSessionInfo := *,      // @sic R5s201387 Baseline Moving sic@
                                                 template ReleaseAssistanceInd p_ReleaseAssistanceInd := *,
                                                 template PortManagementInfoContainer p_PortManagementInfoContainer := *,
                                                 template IPHeaderCompressionConfig p_IPHeaderCompConfig := *,
                                                 template EthernetHeaderCompressConfig p_EthernetHeaderCompConfig := *) return boolean
  {
    var boolean v_Result := false;
    var boolean v_ExistingId := false;
    var NG_NAS_UL_Message_Type v_GSMMsg := p_ReceivedMsgs.PiggybackedPduList[0].Msg;
    var integer i;
    
    // Check NAS Transport msg first
    v_Result := f_Check_NG_ULNASTransport (p_ReceivedMsgs.Msg,
                                           cr_PayloadContainerType(tsc_PayloadContainerESMMsg), // @sic R5w190208 sic@
                                           p_PDU_SessionId,
                                           p_OldPDU_SessionId,
                                           p_RequestType,
                                           p_S_NSSAI,
                                           p_DNN,
                                           p_AdditionalInfo,
                                           p_MA_PDUSessionInfo,   // @sic R5s201387 Baseline Moving sic@
                                           p_ReleaseAssistanceInd);
    if (not v_Result) {
      return v_Result;
    } else if (match (v_GSMMsg, cr_NG_PDU_SESSION_MODIFICATION_REQUEST (cr_NG_PDU_SessionId(-, p_PDU_SessionId), // @sic R5s200564 sic@
                                                                        p_PTI,
                                                                        p_NG_UE_SM_Cap,
                                                                        p_Cause,
                                                                        p_MaxNumPacketFilters,
                                                                        p_AlwaysOnPDUSessionReq,
                                                                        p_IntegrityProtMaxDataRate,
                                                                        p_QoS_Rules,
                                                                        p_QoSFlowDescr,
                                                                        p_MappedEPSBearerContexts,
                                                                        p_ExtdPCO,
                                                                        p_PortManagementInfoContainer,  // @sic R5s201387 Baseline Moving sic@
                                                                        p_IPHeaderCompConfig,
                                                                        p_EthernetHeaderCompConfig))) {
      v_Result := true;
    } else {  // @sic R5s190944 sic@
      v_Result := false;
    }
    // Default message contents specifies this must match an existing session establishment id
    for (i := 0; i < lengthof(p_PDUSessionGlobalInfo); i := i+1) {
     if (match (v_GSMMsg.pdu_Session_Modification_Request.pduSessionId.sessionId, p_PDUSessionGlobalInfo[i].SessionId)) { // @sic R5s200564 sic@
       v_ExistingId := true;
       p_GSM_MobilityInfo := p_PDUSessionGlobalInfo[i];
     }
    }
    if (not v_ExistingId) {
      v_Result := v_ExistingId;
    }
    
    // Store info for this modification
    p_GSM_MobilityInfo.SessionId := v_GSMMsg.pdu_Session_Modification_Request.pduSessionId.sessionId; // @sic R5s200564 sic@
    p_GSM_MobilityInfo.PTI := v_GSMMsg.pdu_Session_Modification_Request.procedureTransactionIdentifier;
    p_GSM_MobilityInfo.SessionType := omit;
    p_GSM_MobilityInfo.EPS_Bearer := omit; // @sic R5s201533 sic@
    p_GSM_MobilityInfo.PdnIndex := omit; // @sic R5s201533 sic@
    
    if (ispresent (p_ReceivedMsgs.Msg.ul_Nas_Transport.dnn)) {
      p_GSM_MobilityInfo.DNN := p_ReceivedMsgs.Msg.ul_Nas_Transport.dnn;
    } /* REMOVED else {
      p_GSM_MobilityInfo.DNN := omit;
    }
    p_GSM_MobilityInfo.PDUType := f_GetPDNTypeFromDNN(p_GSM_MobilityInfo.DNN);  // @sic R5-206296 sic@ */
    if (ispresent (v_GSMMsg.pdu_Session_Modification_Request.alwaysOnPDUSessionReq)) {
      p_GSM_MobilityInfo.AlwaysOn := v_GSMMsg.pdu_Session_Modification_Request.alwaysOnPDUSessionReq;
    } /* REMOVED  else {
      p_GSM_MobilityInfo.AlwaysOn := omit;
    } */
    if (ispresent (v_GSMMsg.pdu_Session_Modification_Request.extdProtocolConfigurationOptions)) {
      p_GSM_MobilityInfo.PCO := v_GSMMsg.pdu_Session_Modification_Request.extdProtocolConfigurationOptions;
    } /* REMOVED else {
      p_GSM_MobilityInfo.PCO := omit;
    } */
    p_GSM_MobilityInfo.NSSAI := omit; // @sic R5s210408 sic@
    return v_Result;
  }
	


1. [bookmark: _Toc100777901][bookmark: _Toc122434494][bookmark: _Toc295288971][bookmark: _Toc325725666][bookmark: _Toc66864323][bookmark: _Toc84011981]Change 2
	Function name
	f_NR5GC_IRAT_S1ToN1_SessionModification()

	Reason for change
	1. Current TTCN implementation is setting p_ExpectedNumberOfNewPDUSessions = pc_noOf_PDUsSameConnection + pc_noOf_PDUsNewConnection. According to 38.508-1 table 4.5A.2C.1-1 the value should be based on the PDN pics parameters (used for S1 mode operation) not PDU. Also, the parameter pc_noOf_PDNsSameConnection is defined as : “Number of UE requested PDN connection establishments after ATTACH in a new signaling connection”. This implies that the PDN connection which is made during the initial EPS attach procedure is not counted in these PICs parameters. So the total count of the PDN connections established in S1 mode (and hence the value for  p_ExpectedNumberOfNewPDUSessions) will be “pc_noOf_PDNsSameConnection + pc_noOf_PDNsNewConnection + 1”

2. In the current implementation the index used when retrieving a stored PDU SESSION MODIFICATION REQUEST is hardcoded to 1. This means the 1st received request will always be retrieved at each loop iteration.

3. In the case where a 2nd PDU SESSION MODIFICATION REQUEST has been received and stored while the 1st is being processed, the current function implementation, after sending PDU SESSION MODIFICATION COMMAND for the 1st request, will retrieve the stored PDU SESSION MODIFICATION REQUEST, increment the counter L to 2, and then deciding not to process it because it is waiting for PDU SESSION MODIFICATION COMPLETE. But because the counter L has already been incremented, after receiving PDU SESSION MODIFICATION COMPLETE, the function will then check L=N and so then does not trigger the processing of the 2nd request.

	Summary of change
	1. Correct setting of p_ExpectedNumberOfNewPDUSessions to “pc_noOf_PDNsSameConnection + pc_noOf_PDNsNewConnection+1”

Note : An associated prose CR on 38.508-1 will be raised at RAN5#95.

2. Use the index “L+1” when calling f_NR5GC_MsgInDefault_GetOneNASMessageReq() to retrieve each stored PDU SESSION MODIFICATION REQUEST in turn.

3. Do not check for any stored messages if the variable v_WaitForComplete is set TRUE, since in this case the request will anyway not be processed.

	TTCN module
	NR5GC_IRAT_CommonProcedures.ttcn

	MCC160 Comment
	Accepted subject to prose CR approval



Before Change
	function f_NR5GC_IRAT_S1ToN1_SessionModification(NR_CellId_Type p_NR_CellId,
                                                   EUTRA_NR_PduSessionInfoList_Type p_PduSessionsFromLTE,
                                                   boolean p_FailWhenK_LessN := true,
                                                   NR_RadioBearerId_Type p_SRB := tsc_NR_RbId_SRB1,
                                                   integer p_ExpectedNumberOfNewPDUSessions := (pc_noOf_PDUsSameConnection + pc_noOf_PDUsNewConnection),
                                                   boolean p_DRBsAlreadyConfigured := false) runs on NR5GC_PTC return boolean
  {
    var template (value) PDUSessionInfoList_Type v_FromLTEList;
    var NG_NAS_UL_PduList_Type v_PduSessionModificationRequestList := {};
    var IntegerList_Type v_PDU_SessionIdList := {};
    var GMM_MobilityInfo_Type v_GMM_MobilityInfo := f_NR5GC_MobileInfo_GetGMM_MobilityInfo();
    var GSM_MobilityInfo_Type v_GSM_MobilityInfo;
    var NR_SRB_COMMON_IND v_ReceivedAsp;
    var NG_NAS_UL_Pdu_Type v_NG_NAS_UL_Pdu;
    var NG_NAS_UL_Message_Type v_GSMMsg;
    var integer v_PDU_SessionId;
    var template (value) RadioBearerConfig v_RadioBearerConfig;
    var template(value) CellGroupConfig v_CellGroupConfig;
    var template (value) NR_RadioBearerList_Type v_RadioBearerList;  // list of DRBs to be configured at the SS (SRB2 alredy configured in SS at cell creation/RRC release procedure => not needed here)
    var template (value) SDAP_ConfigurationList_Type v_SDAP_ConfigurationList;
    var boolean v_ReregisterIMS := false;
    var template (value) NG_NAS_DL_Pdu_Type v_NG_NAS_DL_Pdu;
    var integer N := p_ExpectedNumberOfNewPDUSessions;  // number of PDU SESSION MODIFICATION REQUEST messages being expected
    var integer K := 0;   // number of PDU SESSION MODIFICATION REQUEST messages already processed (i.e. the corresponding PDU SESSION MODIFICATION COMMAND and PDU SESSION MODIFICATION COMPLETE have been sent/received)
    var integer L := 0;   // number of PDU SESSION MODIFICATION REQUEST messages being received
    var boolean v_WaitForComplete := false;
    var template (present) NG_UE_SM_Cap v_NG_UE_SM_Cap := ?;                         // according to 38.508-1 Table 4.5A.2C.2.3-1
    var template (present) IntegrityProtMaxDataRate v_IntegrityProtMaxDataRate := ?; // according to 38.508-1 Table 4.5A.2C.2.3-1
    var integer i;
    var SRBs_AlreadyConfigured v_SRBs := SRB1;
    var NR_RadioBearerId_Type v_SRB := p_SRB;  // @sic R5s210672 sic@
    var IntegerList_Type v_TransferredSessions;
    var PDU_PDN_DNN_Type v_PDU_Type := Internet_DNN; // assume this for now @sic R5-206296 sic@
    var boolean v_RxdMsg := false;
    var boolean v_WaitForRRCComplete := false;
    
    timer t_Wait := 10.0; // @sic R5-204330 sic@

    // Build the GSMMobilityInfo for all received PDUs, but don't store it (some may be established, rather than modified)
    if (ispresent(v_GMM_MobilityInfo.SessionStatus)) { // @sic R5-206290 sic@
      v_TransferredSessions := f_NG_ConvertPDU_SessionStatus(valueof(v_GMM_MobilityInfo.SessionStatus));
      for (i := 0; i < lengthof (p_PduSessionsFromLTE); i := i + 1) {
        if (f_IntegerList_Search(v_TransferredSessions, p_PduSessionsFromLTE[i].PDU_SessionId) != -1) {
          // don't store if not in transferred list
          if (p_PduSessionsFromLTE[i].IsIMS) {  // @sic R5-206296 sic@
            v_PDU_Type := IMS_DNN;
          }
          v_FromLTEList[i] := cs_GSM_MobilityInfo(int2oct(p_PduSessionsFromLTE[i].PDU_SessionId, 1), -, v_PDU_Type); // @sic R5s200564 sic@
        }
      }
    }
        
    t_Wait.start;  // @sic R5-204330 sic@
    while ((K < N) or v_WaitForRRCComplete) { // @sic R5-215678 sic@
      if (f_NR5GC_MsgInDefault_GetNumOfNASMessageReq() > L) { // @sic R5s210520, R5s220174 sic@
        // 1st msg was received before entering function
        v_ReceivedAsp := f_NR5GC_MsgInDefault_GetOneNASMessageReq(1); // @sic R5s220174 sic@
        v_RxdMsg := true;
        t_Wait.stop;
      } else {
        alt { // @sic R5-204330 sic@
          [v_WaitForRRCComplete] SRB.receive(car_NR_SRB_RrcPdu_IND(p_NR_CellId,
                                      tsc_NR_RbId_SRB1,
                                      cr_38508_RRCReconfigurationComplete))     // @sic R5s211280 sic@
          {
            v_WaitForRRCComplete := false;
          }
          [] SRB.receive(car_NR_SRB_NasPdu_IND(p_NR_CellId, v_SRB, cr_NG_NAS_IndWithPiggybacking(tsc_SHT_IntegrityProtected_Ciphered))) -> value v_ReceivedAsp // @sic R5s201413, R5s210672 sic@
          {
            v_RxdMsg := true;
            t_Wait.stop;
          }
          [] t_Wait.timeout {
            if (p_FailWhenK_LessN and (K < N)) {
              f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Not enough PDU Session Modification Request messages");
            } else if (not p_FailWhenK_LessN) { // New message not received in time, so just end the loop  @sic R5-204330 sic@
              break;
            }
          }
        }
      }
      if (v_RxdMsg) { // @sic R5s210520 sic@
        v_NG_NAS_UL_Pdu := v_ReceivedAsp.Signalling.Nas[0].Pdu;
        v_GSMMsg := v_NG_NAS_UL_Pdu.PiggybackedPduList[0].Msg;

        if (ischosen(v_GSMMsg.pdu_Session_Modification_Request)) {
          v_PduSessionModificationRequestList[L] := v_NG_NAS_UL_Pdu;
          L := L + 1;
          if (L > N) { // @sic R5-204330 sic@
              f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "L > N");
          }
          if (not v_WaitForComplete and (K < L)) {   // => process the next PDU SESSION MODIFICATION REQUEST
            v_NG_NAS_UL_Pdu := v_PduSessionModificationRequestList[K];

            if (not f_Check_NG_PDUSessionModificationReq(v_GSM_MobilityInfo, valueof(v_FromLTEList), v_NG_NAS_UL_Pdu, -, -, v_NG_UE_SM_Cap, -, -, -, v_IntegrityProtMaxDataRate, -, -, -, -, -, cr_NG_Request_Type('101'B))) {     // v_GSM_MobilityInfo is out parameter of f_Check_NG_PDUSessionModificationReq
              f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Invalid PDU SESSION MODIFICATION REQUEST");
            }
            // ok, this PDU is to be modified, so can now store it in our list
            v_PDU_SessionId := oct2int(v_GSM_MobilityInfo.SessionId); // @sic R5s200564 sic@
            if (f_IntegerList_Search(v_PDU_SessionIdList, v_PDU_SessionId) < 0) {
              v_PDU_SessionIdList[K] := v_PDU_SessionId;
            } else {
              f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "There has been a PDU SESSION MODIFICATION REQUEST with the same session id already");
            }
            for (i := 0; i < N; i := i + 1) {
              if (p_PduSessionsFromLTE[i].PDU_SessionId == oct2int(v_GSM_MobilityInfo.SessionId)) {
     //             f_IP_ChangeDrbMappingNR(IP, p_NR_CellId, cs_IP_QosFlowId(p_PduSessionsFromLTE[i].PDU_SessionId, p_PduSessionsFromLTE[i].QFI), p_PduSessionsFromLTE[i].PdnIndex);
                v_GSM_MobilityInfo.EPS_Bearer := valueof(p_PduSessionsFromLTE[i].EPS_Bearer); // @sic R5s210408 sic@
                v_GSM_MobilityInfo.PdnIndex := valueof(p_PduSessionsFromLTE[i].PdnIndex); // @sic R5s210408 sic@
                v_ReregisterIMS := v_ReregisterIMS or p_PduSessionsFromLTE[i].IsIMS;
                break;
              }
            }
            f_NR5GC_MobileInfo_AddPDUSession(v_GSM_MobilityInfo); // @sic R5s201413, R5s210408 sic@
            // At the moment no DRBs are configured in the prose - so this will only work for HO TCs
            if (v_ReregisterIMS) { // @sic R5-205942 sic@
              // Tell IMS to start waiting for the Register message.
              f_IMS_IPCAN_SendCoOrdMsg(IMS[tsc_Index_IMS1], cms_IPCAN_IMS_ReregistrationStart(NR));
            }
            v_NG_NAS_DL_Pdu := f_Get_NG_PDUSessionModificationCommand(v_GSM_MobilityInfo.SessionId, v_GSM_MobilityInfo);
            f_BuildRRCReconfigParamsNewDRBs (v_RadioBearerConfig,
                                             v_CellGroupConfig,
                                             v_RadioBearerList,
                                             v_SDAP_ConfigurationList,
                                             p_NR_CellId,
                                             {v_GSM_MobilityInfo},
                                             v_SRBs);  // @sic R5-213442 sic@
            if (v_SRB == tsc_NR_RbId_SRB1) { // @sic R5s201654, R5s210672 sic@
              v_SRBs := SRB1_2;
              v_SRB := tsc_NR_RbId_SRB2;
            }
            if (p_DRBsAlreadyConfigured) {  // @sic R5-213442 sic@
              SRB.send(cas_NR_SRB_NasPdu_REQ(p_NR_CellId, v_SRB, -, cs_NG_NAS_RequestWithPiggybacking(tsc_SHT_IntegrityProtected_Ciphered, v_NG_NAS_DL_Pdu))); // @sic R5s201413, R5s210672 sic@
              v_WaitForComplete := true;  // @sic R5-215678 sic@
            } else {                        // @sic R5-213442 sic@
              // Send with a RRC Reconfiguration message
              f_NR_RRCReconfigNewDRBWithSSConfig(p_NR_CellId,
                                                 v_RadioBearerConfig,
                                                 v_RadioBearerList,
                                                 v_CellGroupConfig,
                                                 { cs_NG_NAS_RequestWithPiggybacking(tsc_SHT_IntegrityProtected_Ciphered, v_NG_NAS_DL_Pdu)},
                                                 -, -, false); // @sic R5-215678 sic@
            }
            v_WaitForComplete := true;
            v_WaitForRRCComplete := true; // @sic R5-215678 sic@
            t_Wait.start;  // restart timer to wait for next Request
          }
        }
        else if (v_WaitForComplete and ischosen(v_GSMMsg.pdu_Session_Modification_Complete)) {
          // The v_GSM_MobilityInfo WILL be initialised when this iteration of the loop is processed
          if (not f_Check_NG_PDUSessionModificationComplete(v_NG_NAS_UL_Pdu, v_GSM_MobilityInfo.SessionId, v_GSM_MobilityInfo.PTI)) {
            f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Invalid PDU SESSION MODIFICATION COMPLETE");
          }
          v_WaitForComplete := false;
          K := K + 1;
        }
        else {
          f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Unexpected NAS PDU");
        }

      }
      v_RxdMsg := false; // @sic R5s210520 sic@
    }
    // Timer will have been started again in loop, but loop is now finished, so need to stop timer
    t_Wait.stop;
    
    return v_ReregisterIMS;
  }



After Change
	function f_NR5GC_IRAT_S1ToN1_SessionModification(NR_CellId_Type p_NR_CellId,
                                                   EUTRA_NR_PduSessionInfoList_Type p_PduSessionsFromLTE,
                                                   boolean p_FailWhenK_LessN := true,
                                                   NR_RadioBearerId_Type p_SRB := tsc_NR_RbId_SRB1,
                                                   integer p_ExpectedNumberOfNewPDUSessions := (pc_noOf_PDNsSameConnection + pc_noOf_PDNsNewConnection +1),
                                                   boolean p_DRBsAlreadyConfigured := false) runs on NR5GC_PTC return boolean
  {
    var template (value) PDUSessionInfoList_Type v_FromLTEList;
    var NG_NAS_UL_PduList_Type v_PduSessionModificationRequestList := {};
    var IntegerList_Type v_PDU_SessionIdList := {};
    var GMM_MobilityInfo_Type v_GMM_MobilityInfo := f_NR5GC_MobileInfo_GetGMM_MobilityInfo();
    var GSM_MobilityInfo_Type v_GSM_MobilityInfo;
    var NR_SRB_COMMON_IND v_ReceivedAsp;
    var NG_NAS_UL_Pdu_Type v_NG_NAS_UL_Pdu;
    var NG_NAS_UL_Message_Type v_GSMMsg;
    var integer v_PDU_SessionId;
    var template (value) RadioBearerConfig v_RadioBearerConfig;
    var template(value) CellGroupConfig v_CellGroupConfig;
    var template (value) NR_RadioBearerList_Type v_RadioBearerList;  // list of DRBs to be configured at the SS (SRB2 alredy configured in SS at cell creation/RRC release procedure => not needed here)
    var template (value) SDAP_ConfigurationList_Type v_SDAP_ConfigurationList;
    var boolean v_ReregisterIMS := false;
    var template (value) NG_NAS_DL_Pdu_Type v_NG_NAS_DL_Pdu;
    var integer N := p_ExpectedNumberOfNewPDUSessions;  // number of PDU SESSION MODIFICATION REQUEST messages being expected
    var integer K := 0;   // number of PDU SESSION MODIFICATION REQUEST messages already processed (i.e. the corresponding PDU SESSION MODIFICATION COMMAND and PDU SESSION MODIFICATION COMPLETE have been sent/received)
    var integer L := 0;   // number of PDU SESSION MODIFICATION REQUEST messages being received
    var boolean v_WaitForComplete := false;
    var template (present) NG_UE_SM_Cap v_NG_UE_SM_Cap := ?;                         // according to 38.508-1 Table 4.5A.2C.2.3-1
    var template (present) IntegrityProtMaxDataRate v_IntegrityProtMaxDataRate := ?; // according to 38.508-1 Table 4.5A.2C.2.3-1
    var integer i;
    var SRBs_AlreadyConfigured v_SRBs := SRB1;
    var NR_RadioBearerId_Type v_SRB := p_SRB;  // @sic R5s210672 sic@
    var IntegerList_Type v_TransferredSessions;
    var PDU_PDN_DNN_Type v_PDU_Type := Internet_DNN; // assume this for now @sic R5-206296 sic@
    var boolean v_RxdMsg := false;
    var boolean v_WaitForRRCComplete := false;
    
    timer t_Wait := 10.0; // @sic R5-204330 sic@

    // Build the GSMMobilityInfo for all received PDUs, but don't store it (some may be established, rather than modified)
    if (ispresent(v_GMM_MobilityInfo.SessionStatus)) { // @sic R5-206290 sic@
      v_TransferredSessions := f_NG_ConvertPDU_SessionStatus(valueof(v_GMM_MobilityInfo.SessionStatus));
      for (i := 0; i < lengthof (p_PduSessionsFromLTE); i := i + 1) {
        if (f_IntegerList_Search(v_TransferredSessions, p_PduSessionsFromLTE[i].PDU_SessionId) != -1) {
          // don't store if not in transferred list
          if (p_PduSessionsFromLTE[i].IsIMS) {  // @sic R5-206296 sic@
            v_PDU_Type := IMS_DNN;
          }
          v_FromLTEList[i] := cs_GSM_MobilityInfo(int2oct(p_PduSessionsFromLTE[i].PDU_SessionId, 1), -, v_PDU_Type); // @sic R5s200564 sic@
        }
      }
    }
        
    t_Wait.start;  // @sic R5-204330 sic@
    while ((K < N) or v_WaitForRRCComplete) { // @sic R5-215678 sic@
      if (not (v_WaitForComplete) and f_NR5GC_MsgInDefault_GetNumOfNASMessageReq() > L) { // @sic R5s210520, R5s220174 sic@
        // 1st msg was received before entering function
        v_ReceivedAsp := f_NR5GC_MsgInDefault_GetOneNASMessageReq(L+1); // @sic R5s220174 sic@
        v_RxdMsg := true;
        t_Wait.stop;
      } else {
        alt { // @sic R5-204330 sic@
          [v_WaitForRRCComplete] SRB.receive(car_NR_SRB_RrcPdu_IND(p_NR_CellId,
                                      tsc_NR_RbId_SRB1,
                                      cr_38508_RRCReconfigurationComplete))     // @sic R5s211280 sic@
          {
            v_WaitForRRCComplete := false;
          }
          [] SRB.receive(car_NR_SRB_NasPdu_IND(p_NR_CellId, v_SRB, cr_NG_NAS_IndWithPiggybacking(tsc_SHT_IntegrityProtected_Ciphered))) -> value v_ReceivedAsp // @sic R5s201413, R5s210672 sic@
          {
            v_RxdMsg := true;
            t_Wait.stop;
          }
          [] t_Wait.timeout {
            if (p_FailWhenK_LessN and (K < N)) {
              f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Not enough PDU Session Modification Request messages");
            } else if (not p_FailWhenK_LessN) { // New message not received in time, so just end the loop  @sic R5-204330 sic@
              break;
            }
          }
        }
      }
      if (v_RxdMsg) { // @sic R5s210520 sic@
        v_NG_NAS_UL_Pdu := v_ReceivedAsp.Signalling.Nas[0].Pdu;
        v_GSMMsg := v_NG_NAS_UL_Pdu.PiggybackedPduList[0].Msg;

        if (ischosen(v_GSMMsg.pdu_Session_Modification_Request)) {
          v_PduSessionModificationRequestList[L] := v_NG_NAS_UL_Pdu;
          L := L + 1;
          if (L > N) { // @sic R5-204330 sic@
              f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "L > N");
          }
          if (not v_WaitForComplete and (K < L)) {   // => process the next PDU SESSION MODIFICATION REQUEST
            v_NG_NAS_UL_Pdu := v_PduSessionModificationRequestList[K];

            if (not f_Check_NG_PDUSessionModificationReq(v_GSM_MobilityInfo, valueof(v_FromLTEList), v_NG_NAS_UL_Pdu, -, -, v_NG_UE_SM_Cap, -, -, -, v_IntegrityProtMaxDataRate, -, -, -, -, -, cr_NG_Request_Type('101'B))) {     // v_GSM_MobilityInfo is out parameter of f_Check_NG_PDUSessionModificationReq
              f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Invalid PDU SESSION MODIFICATION REQUEST");
            }
            // ok, this PDU is to be modified, so can now store it in our list
            v_PDU_SessionId := oct2int(v_GSM_MobilityInfo.SessionId); // @sic R5s200564 sic@
            if (f_IntegerList_Search(v_PDU_SessionIdList, v_PDU_SessionId) < 0) {
              v_PDU_SessionIdList[K] := v_PDU_SessionId;
            } else {
              f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "There has been a PDU SESSION MODIFICATION REQUEST with the same session id already");
            }
            for (i := 0; i < N; i := i + 1) {
              if (p_PduSessionsFromLTE[i].PDU_SessionId == oct2int(v_GSM_MobilityInfo.SessionId)) {
     //             f_IP_ChangeDrbMappingNR(IP, p_NR_CellId, cs_IP_QosFlowId(p_PduSessionsFromLTE[i].PDU_SessionId, p_PduSessionsFromLTE[i].QFI), p_PduSessionsFromLTE[i].PdnIndex);
                v_GSM_MobilityInfo.EPS_Bearer := valueof(p_PduSessionsFromLTE[i].EPS_Bearer); // @sic R5s210408 sic@
                v_GSM_MobilityInfo.PdnIndex := valueof(p_PduSessionsFromLTE[i].PdnIndex); // @sic R5s210408 sic@
                v_ReregisterIMS := v_ReregisterIMS or p_PduSessionsFromLTE[i].IsIMS;
                break;
              }
            }
            f_NR5GC_MobileInfo_AddPDUSession(v_GSM_MobilityInfo); // @sic R5s201413, R5s210408 sic@
            // At the moment no DRBs are configured in the prose - so this will only work for HO TCs
            if (v_ReregisterIMS) { // @sic R5-205942 sic@
              // Tell IMS to start waiting for the Register message.
              f_IMS_IPCAN_SendCoOrdMsg(IMS[tsc_Index_IMS1], cms_IPCAN_IMS_ReregistrationStart(NR));
            }
            v_NG_NAS_DL_Pdu := f_Get_NG_PDUSessionModificationCommand(v_GSM_MobilityInfo.SessionId, v_GSM_MobilityInfo);
            f_BuildRRCReconfigParamsNewDRBs (v_RadioBearerConfig,
                                             v_CellGroupConfig,
                                             v_RadioBearerList,
                                             v_SDAP_ConfigurationList,
                                             p_NR_CellId,
                                             {v_GSM_MobilityInfo},
                                             v_SRBs);  // @sic R5-213442 sic@
            if (v_SRB == tsc_NR_RbId_SRB1) { // @sic R5s201654, R5s210672 sic@
              v_SRBs := SRB1_2;
              v_SRB := tsc_NR_RbId_SRB2;
            }
            if (p_DRBsAlreadyConfigured) {  // @sic R5-213442 sic@
              SRB.send(cas_NR_SRB_NasPdu_REQ(p_NR_CellId, v_SRB, -, cs_NG_NAS_RequestWithPiggybacking(tsc_SHT_IntegrityProtected_Ciphered, v_NG_NAS_DL_Pdu))); // @sic R5s201413, R5s210672 sic@
              v_WaitForComplete := true;  // @sic R5-215678 sic@
            } else {                        // @sic R5-213442 sic@
              // Send with a RRC Reconfiguration message
              f_NR_RRCReconfigNewDRBWithSSConfig(p_NR_CellId,
                                                 v_RadioBearerConfig,
                                                 v_RadioBearerList,
                                                 v_CellGroupConfig,
                                                 { cs_NG_NAS_RequestWithPiggybacking(tsc_SHT_IntegrityProtected_Ciphered, v_NG_NAS_DL_Pdu)},
                                                 -, -, false); // @sic R5-215678 sic@
            }
            v_WaitForComplete := true;
            v_WaitForRRCComplete := true; // @sic R5-215678 sic@
            t_Wait.start;  // restart timer to wait for next Request
          }
        }
        else if (v_WaitForComplete and ischosen(v_GSMMsg.pdu_Session_Modification_Complete)) {
          // The v_GSM_MobilityInfo WILL be initialised when this iteration of the loop is processed
          if (not f_Check_NG_PDUSessionModificationComplete(v_NG_NAS_UL_Pdu, v_GSM_MobilityInfo.SessionId, v_GSM_MobilityInfo.PTI)) {
            f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Invalid PDU SESSION MODIFICATION COMPLETE");
          }
          v_WaitForComplete := false;
          K := K + 1;
        }
        else {
          f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Unexpected NAS PDU");
        }

      }
      v_RxdMsg := false; // @sic R5s210520 sic@
    }
    // Timer will have been started again in loop, but loop is now finished, so need to stop timer
    t_Wait.stop;
    
    return v_ReregisterIMS;
  }




